
ABSTRACT 
In certain circumstances it is desirable to retain the file datetime stamp when moving or copying a file from one location 
to another.  This is a trivial manner when performing such functions under one operating system because the COPY 
and/or MOVE commands typically do not alter the datetime stamp.  But what about moving or copying files between 
different operating systems?  The standard file transfer protocol (FTP) does not give an option to preserve datetime 
stamps, but many vendors have created FTP software with this feature.  This short paper will describe how to achieve 
this outcome with SAS code.  The code transfers files between the OpenVMS operating system and Windows using 
SAS/CONNECT.  Windows API functions are then called to set a file’s datetime stamp to match with VMS.  This code 
was used as part of a platform migration project.  With slight modifications, the code should work equally well for Unix-to-
Windows file transfers.  

 

INTRODUCTION 
For a platform migration project, one requirement might be the preservation of file datetime stamps when copying or 
moving files between platforms.  The maintenance of datetime stamps provides for continued ability to make use of this 
important attribute. 
 
To meet the requirements for a migration, tools are chosen or built.  SAS/CONNECT software is one such tool that 
provides file transfer services.  For a SAS migration project, it is an appropriate tool that can handle both ordinary files 
and SAS files.   Built-in functionality allows retention of the datetime stamp carried within a SAS dataset.   Thus, the 
remaining functionality desired is the ability to retain the datetime stamp on the file itself.   This is where the SAS 
MODULE functions and the Windows Application Programming Interface (API) come into play. 
 
The Windows API is a library of routines that exposes the core functionality of the Windows operating system to the 
developer.  It provides access to the guts of Windows.  In the past, only C programmers had the luxury of accessing 
such powerful functions.  However, in recent years, more programming languages provide the ability to call these low-
level functions.  SAS programmers can use the MODULE functions. 
 
Below we explore using this combination of tools to write code that retains file datetime stamps when migrating files from 
OpenVMS to Windows. 

 

SAS/CONNECT BASICS 
SAS/CONNECT software provides the ability for SAS sessions on one machine to communicate with another machine, 
regardless of operating system as long as SAS/CONNECT is installed and licensed on both platforms.  Capabilities 
include remote execution of code and data transfer.  The latter is provided by the PROC DOWNLOAD and PROC 
UPLOAD procedures. 
 
The prime benefit of using SAS/CONNECT in a platform migration is that necessary platform translations, when 
transferring data between two different operating systems, are handled automatically for files including SAS datasets and 
catalogs.  This applies even when the platforms have different versions of SAS.  In our case, OpenVMS was running 8.2 
and the Windows server was running 9.1.3. 

 

WINDOWS API BASICS 

David Johnson wrote an excellent paper for SUGI 30 showing how to use the SAS MODULE functions to access the 
plethora of routines in the Windows API.  And much of the techniques presented here are an application of the concepts 
described in Johnson’s paper. 
 
Basically, the Windows API provides access to the core functionality of Windows.   Some of the Windows API functions 
available are called to search files, create directories, access file attributes, determine the user logged in, ascertain 
amount of free disk space and memory, even to open the CD tray on a computer.  Some of this functionality is already 
provided via SAS functions which are built on top of this API.  Others require the use of MODULE calls. 
 
The API functions are set up with data structures well known to C programmers.  In recent years, other programming 
languages have provided channels to access the API.  In SAS, this involves translating the API function into terms that 
can be understood by SAS.  Johnson describes this as building a prototype that maps a SAS structure to the Windows 
API structure. 
 
There are many reference materials available to guide the developer in finding appropriate API functions and using them.  
The Microsoft Developer Network websites and user-written books help in this area.  As with many things, with a little 
study and curiosity, the programmer can unlock many possibilities. 

 

STRUCTURE OF THE SAS PROGRAM 
Here we begin to describe the flow of a SAS program that transfers files from OpenVMS to Windows and retains the file 
datetime stamp. 
 
The first thing is to set up preliminaries such as designating the VMS source machine to which we connect: 
 

OPTIONS COMAMID=TCP REMOTE=VMSNAME ; 
filename rlink "C:\Program Files\SAS\SAS 9.1\connect\saslink\tcpvms.scr" ; 

 

DETERMINING FILE DATETIME STAMP 
We use the SAS VMS-specific FILEATTR function to determine a file’s creation date on VMS.  For other platforms, one 
might use the PIPE feature of the FILENAME statement to read a directory listing and then parse the file datetimes. 
 

cdt = fileattr (vmsfile,"CDT") ; 
 
*** add leading zero to dates, remove hyphens, and add colon between date and time 
*** ; 
if substr(cdt,1,1) = " " then substr(cdt,1,1) = "0" ; 
cdt = compress(cdt,"-") ; 
do i = 1 to length (cdt) ; 
   if substr(cdt,i,1) = " " then substr(cdt,i,1) = ":" ; 
end ; 
 
*** create a SAS datetime value *** ; 
cdt_n = input (cdt, datetime25.) ; 
format cdt_n datetime25. ; 

 

TRANSFERRING ASCII OR BINARY FILES 
PROC DOWNLOAD is used to transfer ASCII or binary files.  For ASCII files, we use: 
 

rsubmit ; 
proc download infile="[vmspath]vmsfile.ext" outfile="winpath\winfile.ext" status=no 
; 
run ; 
endrsubmit ; 

 

PostersSUGI 31

 



3 

For binary files, the BINARY option is used: 
 

rsubmit ; 
proc download infile="[vmspath]vmsfile.ext" outfile="winpath\winfile.ext" binary 
status=no ; 
run ; 
endrsubmit ; 

 

SAS DATASETS 
There are two steps to preserve SAS dataset dates.  First and foremost, using the DATECOPY option in PROC 
DOWNLOAD preserves the date within the SAS dataset. 
 

rsubmit ; 
proc download data=remotelb.sasdsn out=locallb.sasdn datecopy status=no ; 
run ; 
endrsubmit ; 

 
In the second step, to retain the SAS dataset’s external date, there are two approaches.  We can use PROC 
CONTENTS and output the dataset’s internal date for reference in later code.  Or the approach that we chose to use 
was to get the file system’s date using the same FILEATTR method for determining datetimes for other files.   In either 
method, in order to process a list of files at the same time, a dataset is created with variables describing the physical 
name of the file and the datetime stamp. 
 
The dataset of files and their time stamps is copied to the Windows platform using a PROC DOWNLOAD step. 
 
To access the API, a text file containing prototype definitions must be made and assigned to the fileref SASCBTBL via a 
FILENAME statement.  The following code is an excerpt from the SASCBTBL file.  It is the definition of a Windows API 
function that is understandable to SAS.  In particular, this code sets the datetime stamp on a Windows file.  The 
Windows function is named “SetFileTime” and the expected arguments are defined with SAS equivalents. 
 

ROUTINE SetFileTime 
        minarg=4 
        maxarg=4 
        stackpop=called 
        module=Kernel32 
        returns=long; 
 
  arg  1 num  update  byvalue format=pib4.;      * HANDLE  hFindFile  // file search 
handle ; 
  arg  2 num  update          format=pib8.;      * CONST FILETIME *  lpCreationTime,  
// pointer to creation file time ; 
  arg  3 num  update          format=pib8.;      * CONST FILETIME *  lpModifiedTime,  
// pointer to creation file time ; 
  arg  4 num  update          format=pib8.;      * CONST FILETIME *  lpAccessedTime,  
// pointer to creation file time ; 

 
The following code (adapted from Johnson) shows the use of the MODULE function to set the datetime stamp for the 
files contained in our dataset. 
 

Data SETTIME( Keep = filespec CREATED1 ACCESSE1 WRITTEN1); 
   Set files; 
   CREATED = .; 
   WRITTEN = .; 
   ACCESSED = .; 
   CREATED1 = .; 
   WRITTEN1 = .; 

   ACCESSE1 = .; 
   MSECONDS = 0; 
   SECONDS = int(second(cdt_n)); 
   *** VMS time includes a milli-second component that messes things up because 59.x 
would round up to 60 when x=5 or greater and then Windows doesn.t know what to do 
with the number and thus generates a blank date *** ; 
   MINUTE = Minute( cdt_n); 
   HOUR = Hour( cdt_n); 
   DAY = DAY( DatePart( cdt_n) ); 
   DAYOFWK = WeekDay( DatePart( cdt_n) ); 
   MONTH = MONTH( DatePart( cdt_n) ); 
   YEAR = YEAR( DatePart( cdt_n) ); 
   RC = ModuleN( 'SystemTimeToFileTime', YEAR, MONTH, DAYOFWK, 
      DAY, HOUR, MINUTE, SECONDS, MSECONDS, CREATED); 
   RC = ModuleN( 'LocalFileTimeToFileTime', CREATED, CREATED1); 
 
   *** Modification date is set to same date as created date *** ; 
   RC = ModuleN( 'SystemTimeToFileTime', YEAR, MONTH, DAYOFWK, 
      DAY, HOUR, MINUTE, SECONDS, MSECONDS, WRITTEN); 
   RC = ModuleN( 'LocalFileTimeToFileTime', WRITTEN, WRITTEN1); 
 
   *** Last accessed date is set to same date as created date *** ; 
   RC = ModuleN( 'SystemTimeToFileTime', YEAR, MONTH, DAYOFWK, 
      DAY, HOUR, MINUTE, SECONDS, MSECONDS, ACCESSED); 
   RC = ModuleN( 'LocalFileTimeToFileTime', ACCESSED, ACCESSE1); 
 
   Format CREATED ACCESSED WRITTEN CREATED1 ACCESSE1 WRITTEN1 32.; 
Run; 
 
Data _null_ ; 
   Length PATH $200; 
   Set settime; 
   PATH = filespec ; *** make sure the file is specified with the full path ; 
   RC = ModuleN( 'SetLastError', 0); 
   HANDLE = ModuleN( 'CreateFileA', PATH, 2**30, 0, 0, 3, 0, 0); 
   If HANDLE >= 1 Then Do; 
   RC = ModuleN( 'SetFileTime', HANDLE, CREATED1, ACCESSE1, WRITTEN1); 
   RC = Modulen( 'CloseHandle', HANDLE); 
   End; 
Run; 

 

ISSUES TO CONSIDER 
You will notice in the code creating the SETTIME dataset that we have rounded off the seconds part of the OpenVMS 
datetime: 
 

SECONDS = int(second(cdt_n)); 
 
This is because the SAS “SECOND” function returns both seconds and milliseconds.  This has some unintended 
consequences since the milliseconds are rounded upward if the INT function is not used.  As an example, 59.41 rounds 
up to 60.  However, the Windows API function will not recognize this because the seconds value should be 0-59.  This 
results in the file having a blank datetime stamp, which looks odd in the File Explorer.  Since millisecond data was not too 
significant, we decided to ignore it. 
 
A second issue to consider is the translation to user time zone.  Windows actually stores file datetimes in Greenwich 
Mean Time (GMT).  When displaying the datetime, an offset is applied based on the time zone that is configured for the 

computer.  So our SAS code must take into account the translation of our time value to GMT.  This is done using the 
Windows API function LocalFileTimeToFileTime .  This is illustrated by the following code. 
 

RC = ModuleN( 'LocalFileTimeToFileTime', CREATED, CREATED1); 

 

CONCLUSION 
Being able to access the wide variety of functions offered by the Windows operating system gives added power to the 
SAS programmer.  Among the applications for using the Windows API is the ability to preserve file datetime stamps 
when transferring files from a different operating system onto Windows.  This paper has demonstrated an approach 
used by the author as part of a platform migration project.  SAS/CONNECT offers the basic file transfer capabilities to 
copy files from OpenVMS to Windows.  The Windows API is accessed to preserve the file datetime stamp.  The benefit 
to users is that this important piece of information is not lost during the migration exercise. 

